

Vložit obrázek do listu není nijak složité. Lze využít schránku nebo kartu Vložení / Obrázek. Zarovnání do buňky či oblasti buněk (lépe řečeno přichycení k mřížce) zajistí držení klávesy ALT během přesunu či změně velikosti obrázku. V praxi narážím na dva požadavky – měnit obrázek na základě výběru buňky (obsah udává zdroj obrázku) a přizpůsobovat obrázky buňkám bez zásahu uživatele (nejednotné velikosti obrázků pro produkty, logotypy, ...).



Obrázky pro vložení

Provázanost obrázku na obsah buňky je velmi často řešena komentářem buňky. My tuto možnost ponecháme stranou ze dvou důvodů. Zobrazení komentáře nelze nikdy spolehlivě uřídit (jsou schopny se vykreslit mimo obrazovku), navíc se běžně netisknou. Jednoduše vyhradíme místo na listu pro obrázky, jeho viditelnost zajistíme příčkami a obsah budeme měnit na základě události Worksheet_Change.



Vybraná položka definuje vložený obrázek

Nejprve startovací procedura pro případ, že načtení obrázku budete chtít realizovat s pomocí tlačítka a ne události.

```
1 Sub TestVlozitObrazek()  
2  
3 Dim rngOblastObrazek As Range  
4 Dim strCestaSouborObrazek As String  
5  
6 'definice oblasti pro vložení obrázku  
7 Set rngOblastObrazek = Worksheets("List1").Range("B2:F10")  
8  
9 'zdroj obrázku  
10 strCestaSouborObrazek = ThisWorkbook.Path & "\obrazek1.jpg"  
11 'strCestaSouborObrazek = ThisWorkbook.Path & "\obrazek2.jpg"  
12 'strCestaSouborObrazek = ThisWorkbook.Path & "\obrazek3.jpg"  
13  
14 'vymazání případných původních obrázků v oblasti  
15 Call OblastSmazatObjekty(rngOblastObrazek)  
16  
17 'vložení obrázku do oblasti  
18 'vycentrování v obou směrech a nevynucené přizpůsobení  
19 'tj. větší obrázky se zmenší, menší obrázky se nezvětší  
20 Call VlozitObrazek(strCestaSouborObrazek, rngOblastObrazek, True, _  
21 True, False)  
22  
23 End Sub
```

Následuje výpis nejdůležitějších obslužných procedur. Parametry pro vycentrování jsou volitelné a stejně tak zvládá přizpůsobení menších obrázků oblasti buněk.

```

1  Sub VlozitObrazek(ByVal strSouborObrazek As String, ByVal rngOblastVlozeni As _
2  Range, Optional ByVal bNaStredVodorovne As Boolean = False, Optional ByVal _
3  bNaStredSvisle As Boolean = False, Optional ByVal bZvetsitMensi As Boolean = _
4  False)
5
6  Dim objObrazek As Object
7  Dim dOblastShora As Double
8  Dim dOblastZleva As Double
9  Dim dOblastSirka As Double
10 Dim dOblastVyska As Double
11 Dim dObrazekSirka As Double
12 Dim dObrazekVyska As Double
13 Dim dPomerSirky As Double
14 Dim dPomerVysky As Double
15 Dim dPomerMax As Double
16
17 'zamezení překreslování obrazovky
18 Application.ScreenUpdating = False
19
20 'vložení obrázku
21 Set objObrazek = ActiveSheet.Pictures.Insert(strSouborObrazek)
22
23 'rozměry oblasti pro vložení
24 With rngOblastVlozeni
25 dOblastShora = .Top
26 dOblastZleva = .Left
27 dOblastSirka = .Width
28 dOblastVyska = .Height
29 End With
30
31 'původní rozměry obrázku
32 With objObrazek
33 dObrazekSirka = .Width
34 dObrazekVyska = .Height
35 End With
36
37 'maximální poměr (převrácená hodnota měřítka)
38 dPomerSirky = dObrazekSirka / dOblastSirka
39 dPomerVysky = dObrazekVyska / dOblastVyska
40 dPomerMax = WorksheetFunction.Max(dPomerSirky, dPomerVysky)
41
42 'je potřeba obrázek zmenšit nebo je požadováno
43 'zvětšení malých obrázků do velikosti oblasti?
44 'poměr stran zachován vždy
45 If (dPomerMax > 1) Or (bZvetsitMensi = True) Then
46 'zmenšení (zvětšení)
47 dSirka = dObrazekSirka / dPomerMax
48 dVyska = dObrazekVyska / dPomerMax
49 Else
50 'ponechání rozměrů
51 dSirka = dObrazekSirka
52 dVyska = dObrazekVyska
53 End If
54
55 dShora = dOblastShora
56 dZleva = dOblastZleva
57
58 'vodorovné vycentrování?
59 If bNaStredVodorovne Then
60 dZleva = dZleva + dOblastSirka / 2 - dSirka / 2
61 End If
62
63 'svislé vycentrování?
64 If bNaStredSvisle Then
65 dShora = dShora + dOblastVyska / 2 - dVyska / 2
66 End If
67
68 'nastavení obrázku
69 With objObrazek
70 .Top = dShora
71 .Left = dZleva
72 .Width = dSirka
73 .Height = dVyska
74 End With
75
76 'odstranění proměnné z paměti
77 Set objObrazek = Nothing
78
79 'překreslení obrazovky
80 Application.ScreenUpdating = True
81
82 End Sub

```

Další řádky kódu VBA řeší odstranění obrázku z oblasti buněk (před načtením nového).

1	Sub OblastSmazatObjekty(ByVal rngOblast As Range)
2	
3	Dim shpObjekt As Shape
4	
5	With rngOblast.Parent
6	
7	'pro každý objekt kolekce Shapes na listu
8	For Each shpObjekt In .Shapes
9	
10	'jestliže horní levý roh objektu leží v oblasti
11	If Not Application.Intersect(shpObjekt.TopLeftCell, rngOblast) Is _
12	Nothing Then
13	'a je-li objekt typu obrázek
14	If (shpObjekt. Type = msoPicture) Or (shpObjekt. Type = _
15	msoLinkedPicture) Then
16	'odstranění obrázku
17	shpObjekt.Delete
18	End If
19	End If
20	
21	Next shpObjekt
22	
23	End With
24	
25	End Sub

Jestliže chcete aplikovat změnu na základě události Worksheet_Change, pak stačí drobná obměna startovací procedury. Kód je umístěn v modulu listu.

```

1 Private Sub Worksheet_SelectionChange(ByVal Target As Range)
2
3 Dim rngOblastObrazek As Range
4 Dim rngOblastTest As Range
5
6 Dim strCestaSouborObrazek As String
7
8 'oblast s názvy souborů
9 Set rngOblastTest = Range("B13:F15")
10
11 If Union(rngOblastTest, Target).Address = rngOblastTest.Address Then
12
13 'definice oblasti pro vložení obrázku
14 Set rngOblastObrazek = Range("B2:F10")
15
16 'zdroj obrázku
17 strCestaSouborObrazek = ThisWorkbook.Path & "" & Target.Cells(1).Text
18
19 'vymazání případných původních obrázků v oblasti
20 Call OblastSmazatObjekty(rngOblastObrazek)
21
22 'vložení obrázku do oblasti
23 'vycentrování v obou směrech a nevynucené přizpůsobení
24 'tj. větší obrázky se zmenší, menší obrázky se nezvětší
25 Call VlozitObrazek(strCestaSouborObrazek, rngOblastObrazek, True, True, _
26 False)
27
28 End If
29
30 End Sub

```

Na tomto místě předkládám ještě jednu možnost k zamyšlení. Tou je využití ovládacího prvku Image a jeho vlastností Picture a PictureSizeMode. Načítání obrázku do vlastnosti Picture zprostředkuje metoda LoadPicture. Pro PictureSizeMode doporučuji nastavení na 3 - fmPictureSizeModeZoom (malé obrázky se bohužel budou přizpůsobovat, i když v měřítku). Užití prvku vás zbavuje závislosti na mřížce listu.



Ovládací prvek Image

Událostní procedura pak může vypadat takto:

```
1 Private Sub Worksheet_SelectionChange(ByVal Target As Range)
2
3 Dim rngOblastTest As Range
4
5 'oblast s názvy souborů
6 Set rngOblastTest = Range("B13:F15")
7
8 If Union(rngOblastTest, Target).Address = rngOblastTest.Address Then
9
10 'načtení obrázku do prvku Image1
11 Me.Image1.Picture = LoadPicture(ThisWorkbook.Path & "" & Target.Cells(1).Text)
12
13 End If
14
15 End Sub
```

Příloha:

[excel_vlozeni_obrazku.zip](#)